

## ESTIMAÇÃO DE ESTADO EM SISTEMAS ELÉTRICOS DE POTÊNCIA – UM ESTUDO DE CASO USANDO A LINGUAGEM AMPL



<https://doi.org/10.22533/at.ed.6541125260211>

Data da submissão: 12/12/2025

Data de aceite: 19/12/2025

**Hugo Andrés Ruiz Flórez**

Universidade Tecnológica Federal do  
Paraná - UTFPR  
Medianeira – PR

<http://lattes.cnpq.br/1231733785708599>

**Gloria Patricia Lopez Sepulveda**

Universidade Tecnológica Federal do  
Paraná - UTFPR  
São Miguel do Iguaçu – PR  
<http://lattes.cnpq.br/8402032745107959>

**RESUMO:** Convencionalmente, o problema de Estimação de Estado em Sistemas Elétricos de Potência é resolvido de forma iterativa por meio do método dos Mínimos Quadrados Ponderados (MQP). Embora o método MQP seja amplamente consolidado e utilizado, existe um fator limitante que pode afetar a sua aplicação prática, uma vez que, a implementação computacional é complexa, o que exige um conhecimento avançado de programação. Para contornar essa dificuldade, alternativas como softwares comerciais ou códigos disponibilizados na internet são frequentemente considerados. Entretanto, essas abordagens apresentam limitações importantes, visto que, os softwares

comerciais geralmente não são gratuitos, e os códigos disponíveis on-line nem sempre oferecem a confiabilidade necessária para realizar pesquisas acadêmicas. Diante desse cenário, torna-se relevante dispor de uma ferramenta confiável, acessível e de fácil implementação que permita resolver o problema de Estimação de Estado de maneira confiável. Neste trabalho, propõe-se representar o Estimador de Estado MQP como um modelo matemático de otimização não linear, implementado na linguagem de modelagem matemática AMPL. Esta ferramenta computacional possui uma licença estudantil que permite o uso gratuito de diversos solvers comerciais, o que facilita a realização de pesquisas acadêmicas. O modelo matemático implementado na linguagem AMPL foi testado através de um sistema-teste e solucionado com o solver Knitro. Por fim, os resultados mostram que é possível solucionar satisfatoriamente o problema de Estimação de Estado através da linguagem AMPL.

**PALAVRAS-CHAVE:** Estimação de Estado, Sistemas Elétricos de Potência, Otimização Matemática, AMPL, Solver Knitro.

# POWER SYSTEM STATE ESTIMATION – A CASE STUDY USING THE AMPL LANGUAGE

**ABSTRACT:** Conventionally, the State Estimation problem in Electric Power Systems is solved iteratively using the Weighted Least Squares (WLS) method. Although WLS is a well-established and widely used approach, its practical applicability may be limited by the complexity of its computational implementation, which often requires advanced programming skills. To overcome this challenge, alternatives such as commercial software packages or publicly available code are commonly considered. However, these options present notable limitations: commercial software is typically not free, and publicly available code does not always provide the reliability required for academic research. In this context, having a trustworthy, accessible, and easy-to-implement tool for solving the State Estimation problem becomes particularly valuable. In this work, we propose formulating the WLS State Estimator as a nonlinear optimization model implemented in the AMPL mathematical modeling language. AMPL offers a student license that grants free access to several commercial solvers, facilitating academic research. The proposed mathematical model was tested on a benchmark system and solved using the Knitro solver. The results demonstrate that the State Estimation problem can be successfully addressed through the AMPL modeling framework.

**KEY-WORDS:** State Estimation, Electrical Power Systems, Mathematical Optimization, AMPL Language, Knitro Solver.

## INTRODUCÃO

O ponto de operação de um Sistema de Potência Elétrica (SEP) é determinado através do cálculo das variáveis de estado (módulos e os ângulos das tensões nos barramentos). Logo, uma vez calculado o ponto de operação de um SEP, diversos estudos e procedimentos podem ser realizados, tais como ações de controle e segurança, análise de contingências, planejamento da expansão, estudos de carregamento do sistema, análise de regulação de tensão e avaliação da confiabilidade, entre outros (MONTICELLI, 1983).

Determinar corretamente o ponto de operação de um SEP em regime permanente é uma atividade primordial executada pelos Centros de Controle de Energia. Dentre os procedimentos usados para essa finalidade destacam-se os Estimadores de Estado. Este procedimento desempenha um papel importante nos sistemas elétricos modernos, já que, permite que os centros de controle de energia possam tomar decisões de maneira segura e a confiável (EXPÓSITO, 2004).

Tradicionalmente, a estimativa de estado tem sido resolvida por meio do método dos Mínimos Quadrados Ponderados (MQP). Esse método é caracterizado pelo uso de uma formulação nodal, na qual o ponto de operação de um SEP é obtido através de um procedimento iterativo (MONTICELLI, 2000).

Embora este procedimento tenha sido amplamente utilizado por diversos autores, é possível encontrar algumas limitações que dificultam a realização de estudos de Estimação de Estado. Isto é, apesar de existir vários softwares para realizar estudos de Estimação

de Estado, a maioria destes softwares exigem a compra de uma licença. Por outro lado, apesar de alguns códigos de programação do método MQP estarem disponíveis na internet, a sua confiabilidade pode ser questionada, já que, os resultados podem estar afetados por erros. Por sua vez, outro fator limitante está relacionado com o tempo que pode levar a implementação computacional de um problema complexo em uma linguagem de programação específica, tais como C, C++ ou Matlab, entre outras.

Recentemente tem sido explorada uma alternativa de solução que consiste em representar o problema de Estimação de Estado através de um problema de otimização matemática (Florez et. al., 2021). Esta abordagem tem como principal vantagem poder fazer uso de alguns softwares de modelagem matemática amplamente usados nas análises dos SEP, tais como, GAMS, Lingo, Pyomo, AMPL, entre outros.

Neste trabalho, é proposto implementar um modelo matemático do estimador de estado na sua formulação polar na linguagem AMPL. Assim, para esta finalidade, será realizado um estudo de caso que compreende a determinação do ponto de operação de um sistema-teste disponível em Glover (2011). Os resultados obtidos foram comparados com os resultados obtidos com o fluxo de carga usando o software powerworld. De maneira geral, é possível concluir que os resultados obtidos através da ferramenta AMPL permitem representar e resolver o problema de Estimação de Estado em Sistemas Elétricos de Potência de maneira satisfatória.

## LINGUAGEM DE PROGRAMAÇÃO AMPL

O AMPL é uma linguagem de modelagem algébrica de alto nível amplamente utilizada na formulação e resolução de problemas de otimização matemática, especialmente aqueles de grande porte e complexidade matemática elevada (FAVERO, 2013). Desenvolvida por Fourier et al. (1985), esta linguagem surgiu da necessidade de representar os modelos clássicos de otimização de forma clara, estruturada e compatível com diferentes solvers. Entre suas principais características, destaca-se a sintaxe fortemente inspirada na notação matemática tradicional, o que facilita a representação dos modelos e reduz a distância entre a sintaxe matemática e a implementação computacional (OLSZAK, 2018). Como consequência, os pesquisadores podem concentrar os esforços no aprimoramento das formulações matemáticas, ao invés de dedicar tempo excessivo na implementação computacional dos modelos matemáticos (HOLMES, 1992).

Embora o AMPL não seja um software gratuito, seus desenvolvedores disponibilizam uma versão de uso livre, que permite que os usuários resolvam problemas com até 300 variáveis e 300 restrições no caso dos Problemas de Programação Linear, e 500 variáveis e 500 restrições para Problemas de Programação Não Linear. Essa versão limitada, porém bastante funcional, possibilita a realização de estudos acadêmico sem a necessidade de usar a licença comercial completa.

Essas características explicam os motivos pelos quais a ferramenta AMPL vem ganhando espaço nos últimos anos na solução de diferentes problemas de Engenharia. Diante desse contexto, apresentam-se a seguir os elementos fundamentais que devem ser considerados na resolução de um problema de otimização matemática usando a ferramenta AMPL.

## Resolução de modelos matemático no AMPL

Existem diversas maneiras de resolver um modelo matemático no AMPL. A abordagem mais comum envolve a criação de três arquivos de texto, com extensões .mod, .data e .run, que reúnem, respectivamente, a definição do modelo matemático, os dados de entrada e o conjunto de comandos responsáveis pela execução do problema. No entanto, também é possível obter a solução utilizando apenas dois arquivos .dat e .run. Para este último, pode-se adotar como referência o layout mostrado na Figura 1.

```
reset;
model;
# *****
# Declaração dos Conjuntos (set nome_conjunto;)

# *****
# Declaração dos Parâmetros (param nome_parâmetro;)

# *****
# Dados de Entrada (data nome_arquivo;)

# *****
# Declaração das Variáveis (var nome_variável;)

# *****
# Definição da Função objetivo (minimize nome_função;)

# *****
# Definição das Restrições (subject to nome_restrição;)

# *****
# Solução do modelo matemático (option solver nome_solver;)

# *****
# Saída dos resultados
```

Figura 1 – Layout do arquivo .run usado como referência

Fonte: Autoria própria

Conforme ilustrado na Figura 1, o comando reset é utilizado para limpar da memória qualquer informação relacionada a modelos previamente carregados, enquanto que, o comando model ativa o modo de definição de um modelo matemático. Além disso, por motivos de organização e simplicidade, a formulação matemática pode ser estruturada em diferentes etapas e dividida de forma sequencial.

Nas seções seguintes será apresentado em detalhe como o layout anteriormente apresentado pode ser alterado de modo que seja possível obter a solução do modelo matemático de otimização do problema de estimativa de estado em SEP.

## ESTIMAÇÃO DE ESTADO EM SISTEMAS ELÉTRICOS DE POTÊNCIA

A estimativa de estado é um problema de solução única que consiste na obtenção das variáveis de estado, representadas pelo módulo e o ângulo ( $V, \theta$ ) das tensões nodais, a partir das informações coletadas dos diferentes medidores espalhados no SEP (MONTICELLI, 2000).

De modo geral, o problema de estimativa de estado em SEP pode ser representado matematicamente da seguinte forma:

$$\begin{aligned} & \text{minimizar} && J(\hat{x}) = \sum_{i=1}^{nm} W_{ii} \cdot r_i^2 \\ & \text{sujeito a:} && r_i = z_i - h_i(\hat{x}) \quad \forall i \in \Omega M \end{aligned} \tag{1}$$

Donde  $\hat{x}$  corresponde ao vetor das variáveis de estado;  $J(\hat{x})$  é a função dos mínimos quadrados;  $z$  é o vetor das medidas do SEP;  $r$  é o vetor dos resíduos das medidas;  $W$  é a matriz dos pesos atribuídos a cada medida;  $h(\hat{x})$  representa as equações matemáticas usadas nas análises de fluxo de carga; corresponde ao número de medidas e  $\Omega M$  é o conjunto de medidas.

Idealmente, a solução do problema de estimativa de estado mostrado em (1) consiste em determinar os valores de  $\hat{x}$ , de modo que,  $J(x)=0$ . Porém, essa condição na prática não pode ser alcançada devido a alguns fatores, tais como, a precisão dos medidores, problemas de calibração, os ruídos presentes no sistema SCADA (*Supervisory Control and Data Acquisition*), entre outros (EXPÓSITO, 2004).

Assim sendo, para achar a solução do problema de estimativa de estado em SEP mostrado em (1) têm sido proposto vários métodos, dentre os quais destaca-se o método dos MQP.

## Método dos MQP

O modelo matemático mostrado em (1) pode ser expresso em forma matricial através do seguinte problema de otimização irrestrito:

$$\text{minimizar} \quad J(\hat{x}) = [z - h(\hat{x})]^t \cdot W \cdot [z - h(\hat{x})] \quad (2)$$

Assim, para determinar os valores de que satisfazem o problema anterior, o método dos MQP aplica a condição de optimalidade de primeira ordem, de tal forma que:

$$\frac{\partial J(\hat{x})}{\partial \hat{x}} = H^t W (z - h(\hat{x})) = 0 \quad (3)$$

Onde  $H = \partial h(\hat{x}) / \partial \hat{x}$  define-se como a matriz Jacobiana do sistema.

Consequentemente, a solução da equação (3) é alcançada através de um procedimento iterativo que é executado até atingir uma tolerância ( $\max |\Delta \hat{x}^{(k)}| \leq tol$ ), conforme mostrado na seguinte equação matricial:

$$[\hat{x}^{(k+1)}] = [\hat{x}^{(k)}] + [\Delta \hat{x}^{(k)}] = [\hat{x}^{(k)}] + [G^{(k)}]^{-1} \cdot [H^{(k)}]^t \cdot [W] \cdot [r^{(k)}] \quad (4)$$

Onde  $[G]$ ,  $[H]$  e  $[W]$  são as matrizes ganho, Jacobiana e de ponderações das medidas, respectivamente. Por sua vez,  $[\hat{x}]$  e  $[r]$  são os vetores das variáveis de estado e dos resíduos das medidas, respectivamente (Stevenson, 1986).

## MODELO MATEMÁTICO DO PROBLEMA DE ESTIMAÇÃO DE ESTADO

Para poder representar o problema de Estimação de Estado através de um modelo matemático de otimização é necessário definir uma função objetivo, um conjunto de restrições e as variáveis de decisão do modelo. Assim, tomando como referência a formulação matemática do estimador de estado dos MQP é possível usar o seguinte modelo de otimização não-linear para solucionar o problema de Estimação de Estado:

$$\min \quad J(\hat{x}) = \sum_{[m, i, j] \in \Omega L} W_{mij} r_{mij}^2 \quad (5)$$

s.a.

$$r_{mij} = z_{mij} - h(\hat{x})_{mij}, \quad \forall_{m, i, j} \in \Omega M \quad (6)$$

Sendo que, a função objetivo mostrada em (5) representa a função do método dos MQP. Por sua vez, a restrição (6) corresponde ao resíduo da medição  $m$  calculado como a diferença entre o valor medido  $z_{mij}$  e o valor calculado  $h(\hat{x})_{mij}$ . Cabe salientar que para

calcular  $h(\hat{x})_{mij}$  deve ser considerado o tipo de medida ( $m$ ) e a sua localização ( $ij$ ), como mostrado nas equações (7)-(13):

Medições de injeção de potência ativa ( $P_i$ ) e reativa ( $Q_i$ ):

$$h(\hat{x})_{mij} = \sum_{k \in QB} V_i V_k (G_{ik} \cos \theta_{ik} + B_{ik} \sin \theta_{ik}), \quad \forall_{m,i,j} \in \Omega M : m=1 \quad (7)$$

$$h(\hat{x})_{mij} = \sum_{k \in QB} V_i V_k (G_{ik} \sin \theta_{ik} - B_{ik} \cos \theta_{ik}), \quad \forall_{m,i,j} \in \Omega M : m=2 \quad (8)$$

Onde  $G_{ik}$  e  $B_{ik}$  são os elementos da matriz  $Y_{BUS}$ .

- Medições de fluxo de potência ativa ( $P_{ij}$ ) e ( $P_{ji}$ ):

$$h(\hat{x})_{mij} = g_{ij} V_i^2 - V_i V_j (g_{ij} \cos \theta_{ij} + b_{ij} \sin \theta_{ij}), \quad \forall_{m,i,j} \in \Omega M : m=3 \quad (9)$$

$$h(\hat{x})_{mij} = g_{ij} V_j^2 - V_i V_j (g_{ij} \cos \theta_{ij} - b_{ij} \sin \theta_{ij}), \quad \forall_{m,i,j} \in \Omega M : m=3 \quad (10)$$

- Medições de fluxo de potência reativa ( $Q_{ij}$ ) e ( $Q_{ji}$ ):

$$h(\hat{x})_{mij} = -(b_{ij} + b_{ij}^{sh}) V_i^2 + V_i V_j (b_{ij} \cos \theta_{ij} - g_{ij} \sin \theta_{ij}), \quad \forall_{m,i,j} \in \Omega M : m=4 \quad (11)$$

$$h(\hat{x})_{mij} = -(b_{ij} + b_{ij}^{sh}) V_j^2 + V_i V_j (b_{ij} \cos \theta_{ij} + g_{ij} \sin \theta_{ij}), \quad \forall_{m,i,j} \in \Omega M : m=4 \quad (12)$$

- Medições de tensão ( $V_i$ ):

$$h(\hat{x})_{mij} = V_i, \quad \forall_{m,i,j} \in \Omega M : m=5 \quad (13)$$

Ao resolver o modelo matemático anterior, é possível obter os valores de  $V_i$  e  $\theta_i$  em todas as barras do SEP, assim como, o valor de  $J(\hat{x})$  e os resíduos  $r_{mij}$  de cada medição.

## ESTUDO DE CASO PROPOSTO

### Estudo preliminar

Inicialmente, deve ser realizado um estudo de fluxo de carga para poder criar a base de dados do estimador de estado. Portanto, é usado como referência o sistema-teste disponível em (GLOVER, 2011), cujos resultados da análise de fluxo de carga obtidos com o software PowerWorld são apresentados na Figura 2.




Figura 2 – Resultados do fluxo de carga usando o software PowerWorld

Fonte: Autoria própria

Os dados de entrada do sistema-teste necessários para realizar uma análise convencional de fluxo de carga são apresentados nas Tabelas 1 e 2.

$i$	$Tipo$	$P_i^{G_{esp}}$	$Q_i^{G_{esp}}$	$P_i^{D_{esp}}$	$Q_i^{D_{esp}}$	$Q_i^{D_{esp}}$	$\theta_i^{esp}$	$Q_i^{G_{min}}$	$Q_i^{G_{max}}$	$B_i^{sh}$
1	$V\theta$	--	--	0,00	0,00	1,00	0,00	--		0,00
2	$PQ$	0,00	0,00	8,00	2,80	--	--	--		0,00
3	$PV$	5,20	--	0,80	0,40	1,05	--	-2,80	4,00	0,00
4	$PQ$	0,00	0,00	0,00	0,00	--	--	--		0,00
5	$PQ$	0,00	0,00	0,00	0,00	--	--	--		0,00

Tabela 1 – Dados de entrada referente às barras do sistema-teste

Fonte: Modificado de Glover, 2011

$i$	$j$	$r_{ij}$	$x_{ij}$	$b_{ij}$	$t_{ij}$
1	5	0,00150	0,02000	0,00000	1,00000
3	4	0,00075	0,01000	0,00000	1,00000
2	4	0,00900	0,10000	1,72000	1,00000
2	5	1,00450	0,05000	0,88000	1,00000
4	5	0,00225	0,02500	0,44000	1,00000

Tabela 2 – Dados de entrada referente às LT e transformadores do sistema-teste

Fonte: Modificado de Glover, 2011

Os resultados do fluxo de carga são apresentados nas Tabelas 3 e 4.

$i$	$PG(i)$	$QG(i)$	$V(i)$	$\theta(i)$
1	3,9484	1.1428	1,0000	0,0000
2	0,0000	0,0000	0,8338	-22,4064
3	5,2000	3,3748	1,0500	-0,5973
4	0,0000	0,0000	1,0193	-2,8340
5	0,0000	0,0000	0,9743	-4,5479

Tabela 3 – Valores das potências geradas e as tensões dos barramentos

Fonte: Modificado de Glover, 2011

$i$	$j$	$P(i,j)$	$Q(i,j)$	$P(j,i)$	$Q(j,i)$
1	5	3,9484	1,1428	-3,9230	-0,8049
3	4	4,4000	2,9748	-4,3808	-2,7189
2	4	-2,9184	-1,3911	3,0368	1,2154
2	5	-5,0816	-1,4089	5,2566	2,6302
4	5	1,3440	1,5035	-1,3336	-1,8253

Tabela 4 – Valores dos fluxos de potências das linhas de transmissão

Estes resultados serão tomados como referência para a criação da base de dados do modelo matemático do estimador de estado proposto.

### Dados de entrada do estimador de estado

A partir dos resultados apresentados nas Tabelas 3 e 4, define-se o conjunto de medidas que integrará os dados de entrada do estimador de estado, conforme disponibilizado na Tabela 5.

$Med$	$No.$	$Med$	$No.$	$z_i^{med}$	$Med$	$No.$	$TM$	$z_i^{med}$
1		$P_1$		3,9484		7		$P_{2-5}$
2		$P_2$		3,9484		8		$P_{4-2}$
3		$P_3$		4,4000		9		$Q_{2-5}$

4	$Q_1$	1,1428	10	$Q_{4-2}$	1,2154
5	$Q_2$	-2,8000	11	$V_4$	1,0193
6	$Q_3$	2,9748	12	$V_5$	0,9743

Tabela 5 – Conjunto de medidas do sistema-teste

Neste trabalho, assume-se que as medidas de tensão, fluxo de potência e injeções de potência possuem um desvio padrão de 0.04, 0.08 e 0.01.

Considerando as informações anteriores, é usada a seguinte base de dados do sistema-teste usada para determinar o ponto de operação através do estimador de estado proposto conforme apresentado na Figura 3.

```

param: OB:      TB      Bsh:=
      1         1        0.00
      2         0        0.00
      3         0        0.00
      4         0        0.00
      5         0        0.00
;

param: OL: R      X      Ysh2:=
      1 5  0.00150   0.0200  0.0000
      3 4  0.00075   0.0100  0.0000
      2 4  0.00900   0.1000  0.8600
      2 5  0.00450   0.0500  0.4400
      4 5  0.00225   0.0250  0.2200
;

param: OM:      Zmed     dp:=
      1 1  1       3.9484  0.01
      1 2  2      -8.0000  0.01
      1 3  3       4.4000  0.01
      2 1  1       1.1428  0.01
      2 2  2      -2.8000  0.01
      2 3  3       2.9748  0.01
      3 2  5      -5.0816  0.08
      3 4  2       3.0368  0.08
      4 2  5      -1.4089  0.08
      4 4  2       1.2154  0.08
      5 4  4       1.0193  0.04
      5 5  5       0.9743  0.04
;

```

Figura 3 – Dados entrada do sistema-teste

Fonte: Autoria própria

## Definição de conjuntos e parâmetros

Pode-se apreciar na base de dados do sistema-teste que há necessidade de definir alguns parâmetros que fazem referência aos barramentos, às linhas de transmissão e às medidas. Por tal motivo, faz-se necessário definir os conjuntos e parâmetros mostrados na Figura 4.

```
# ****
# Definição dos Conjuntos (set "nome_conjunto")
# ****
set OB;                      #Conjunto de Barras do sistema
set OL within {OB,OB};        #Conjunto de linhas do sistema
set TM = 1..5;                #Conjunto de tipo de medidas do sistema
set OM within {TM,OB,OB};     #Conjunto de medidas

# ****
# Definição dos Parâmetros (param "nome_parâmetro")
# ****
# Parâmetros das barras com valores definidos na base de dados
param TB{OB};
param Bsh{OB};

# Parâmetros das linhas com valores definidos na base de dados
param R{OL};
param X{OL};
param Ysh2{OL};

# Parâmetros das linhas com valores definidos na base de dados
param Zmed{OM};
param dp{OM};                 # desvio padrão

# Parâmetros com valores calculados depois de importar a base de dados
param b{OL};
param q{OL};
param G{OB,OB};
param B{OB,OB};
param W{OM};
```

Figura 4 – Declaração de conjuntos e parâmetros no AMPL

Fonte: Autoria própria

## Definição das variáveis

De uma forma geral, as incógnitas do problema do modelo matemático proposto podem ser definidas como mostrado na Figura 5:

```
# ****
# Definição das Variáveis (var "nome_variável")
# ****

var V{OB};
var Th{OB};
var Zcal{OM};
var r{OM};
```

Figura 5 – Declaração das variáveis do modelo matemático proposto

Fonte: Autoria própria

### Inicialização das variáveis de decisão

As tensões nodais podem ser inicializadas a partir dos dados de entrada como ilustrado na Figura 6.

```
# ...
# Inicialização dos módulos e ângulos das tensões
# ...

let {i in OB} V[i]:=1;
let {i in OB} Th[i]:=0;
```

Figura 6 – Variáveis do problema com valores especificados

Fonte: Autoria própria

### Definição do modelo matemático

O modelo matemático de otimização do estimador de estado proposto pode ser expresso através da função objetivo e as restrições apresentadas na Figura 7.

```

# ****
# Definição da Função objetivo (minimize "nome_função:")
# ****

minimize fo: sum((m,i,j) in OM) W[m,i,j]*r[m,i,j]^2;

# ****
# Definição das Restrições (subject to "nome_restrição")
# ****

subject to r1 { (m,i,j) in OM}: r[m,i,j] = Zmed[m,i,j] - Zcal[m,i,j];

subject to r2 { (m,i,j) in OM:m==1}: Zcal[m,i,j] = V[i]^2*G[i,i] + sum(k in OB:k!=i)
                                         (V[i]*V[k]*G[i,k]*cos(Th[i]-Th[k])+V[i]*V[k]*B[i,k]*sin(Th[i]-Th[k]));

subject to r3 { (m,i,j) in OM:m==2}: Zcal[m,i,j] = -V[i]^2*B[i,i] + sum(k in OB:k!=i)
                                         (V[i]*V[k]*G[i,k]*sin(Th[i]-Th[k])-V[i]*V[k]*B[i,k]*cos(Th[i]-Th[k]));

subject to r4 { (m,i,j) in OM:m==3 && (i,j) in OL}: Zcal[m,i,j] = V[i]^2*g[i,j] -
                                         V[i]*V[j]*(g[i,j]*cos(Th[i]-Th[j])+b[i,j]*sin(Th[i]-Th[j]));

subject to r5 { (m,i,j) in OM:m==3 && (j,i) in OL}: Zcal[m,i,j] = V[i]^2*g[j,i] -
                                         V[j]*V[i]*(g[j,i]*cos(Th[j]-Th[i])-b[j,i]*sin(Th[j]-Th[i]));

subject to r6 { (m,i,j) in OM:m==4 && (i,j) in OL}: Zcal[m,i,j] = -V[i]^2*(b[i,j]+Ysh2[i,j]) +
                                         V[i]*V[j]*(b[i,j]*cos(Th[i]-Th[j])-g[i,j]*sin(Th[i]-Th[j]));

subject to r7 { (m,i,j) in OM:m==4 && (j,i) in OL}: Zcal[m,i,j] = -V[i]^2*(b[j,i]+Ysh2[j,i]) +
                                         V[j]*V[i]*(b[j,i]*cos(Th[j]-Th[i])+g[j,i]*sin(Th[j]-Th[i]));

subject to r8 { (m,i,j) in OM:m==5}: Zcal[m,i,j] = V[i];

```

Figura 7 – Definição da função objetivo e restrições

Fonte: Autoria própria

## Solução do modelo matemático

De acordo com a formulação do modelo matemático de otimização do estimador de estado proposto deve ser escolhido um solver apropriado para resolver problemas de otimização não-linear. Neste trabalho, será usado o solver Knitro. Assim sendo, em primeiro lugar deve ser escrita a instrução que indica o tipo de solver a ser usado e posteriormente, o nome da função objetivo que deve ser resolvida, tal como apreciado na Figura 8.

```

# ****
# Solução do modelo matemático
# ****

option solver knitroampl;
solve fo;

```

Figura 8 – Definição da função objetivo e restrições

Fonte: Autoria própria

## Dados de saída

Na ferramenta AMPL é possível usar a função printf para imprimir os resultados seja no prompt de um sistema operativo ou em um arquivo de texto. Desta forma, na Figura 9 é apresentado o código usado para poder imprimir os resultados no prompt do Windows.

Figura 9 – Código usado para apresentar os resultados

Fonte: Autoria própria

O código anterior permite obter os resultados das variáveis que pertencem aos conjuntos  $\Omega B$  e  $\Omega M$ .

## Resolução do modelo matemático e apresentação dos resultados

Levando em consideração o exposto anteriormente, deve-se verificar que os arquivos necessários para poder realizar a simulação estejam contidos na mesma pasta de acordo com a Figura 10.

Nome	Tipo
AMPL	Aplicativo
cmd	Atalho
EE_polar	Arquivo RUN
knitroampl	Aplicativo
STEE5b	Arquivo DAT

Figura 10 – Arquivos necessários para a resolução do modelo matemático

Fonte: Autoria própria

Onde os arquivos *AMPL* e *knitroampl* correspondem aos aplicativos que permitem executar o ambiente *AMPL* e o solver que será usado para resolver o problema, respectivamente. O arquivo *cmd* corresponde ao prompt do Windows. Por fim, os arquivos

de texto *EE\_polar.run* e *STEE5b.dat* correspondem ao arquivo principal e o arquivo com os dados do sistema-teste, respectivamente.

Por outro lado, para poder obter os resultados do modelo matemático proposto faz-se necessário escrever no prompt: *AMPL.exe EE\_polar.run*. Dessa maneira, o AMPL retornará um resumo da resolução do modelo matemático proposto usando o Knitro (ver figura 11)

```
Problem Characteristics
-----
Objective goal: Minimize
Number of variables: 33
  bounded below: 0
  bounded above: 0
  bounded below and above: 0
  fixed: 0
  free: 33
Number of constraints: 24
  linear equalities: 14
  nonlinear equalities: 10
  linear inequalities: 0
  nonlinear inequalities: 0
  range: 0
Number of nonzeros in Jacobian: 80
Number of nonzeros in Hessian: 38

  Iter      Objective      FeasError      OptError      ||Step||      CGits
-----  -----
    0  0.000000e+000  8.000e+000
    6  1.301118e-007  3.739e-013  1.173e-010  6.402e-007      0

EXIT: Locally optimal solution found.

Final Statistics
-----
Final objective value          = 1.30111843739111e-007
Final feasibility error (abs / rel) = 3.74e-013 / 4.67e-014
Final optimality error (abs / rel) = 1.17e-010 / 1.17e-010
# of iterations                =       6
# of CG iterations              =       0
# of function evaluations       =       7
# of gradient evaluations      =       7
# of Hessian evaluations        =       6
Total program time (secs)      =      0.009 (      0.000 CPU time)
Time spent in evaluations (secs) =      0.000
```

Figura 11 – Informações da resolução do modelo matemático usando o solver Knitro

Fonte: Autoria própria

Como visto na Figura 11 aparecem algumas informações relacionadas com a resolução do modelo matemático, tais como, número de iterações, valor da função objetivo, tempo de cômputo, entre outras. Por outro lado, outras informações fornecidas pelo solver Knitro são mostradas na Figura 12.

```

Locally optimal solution.
objective 1.301118437e-07; feasibility error 3.74e-13
6 iterations; 7 function evaluations

*****
***** Resultados do estimador de estado *****
***** Cálculo das grandezas do SEP *****



| TM | k | m | Zmed    | Zcal    | ri       | Wii        |
|----|---|---|---------|---------|----------|------------|
| 1  | 1 | 1 | 3.9484  | 3.9484  | 0.00000  | 10000.0000 |
| 1  | 2 | 2 | -8.0000 | -8.0000 | -0.00000 | 10000.0000 |
| 1  | 3 | 3 | 4.4000  | 4.4000  | -0.00000 | 10000.0000 |
| 2  | 1 | 1 | 1.1428  | 1.1428  | 0.00000  | 10000.0000 |
| 2  | 2 | 2 | -2.8000 | -2.8000 | -0.00000 | 10000.0000 |
| 2  | 3 | 3 | 2.9748  | 2.9748  | 0.00000  | 10000.0000 |
| 3  | 2 | 5 | -5.0816 | -5.0816 | 0.00001  | 156.2500   |
| 3  | 4 | 2 | 3.0368  | 3.0368  | -0.00001 | 156.2500   |
| 4  | 2 | 5 | -1.4089 | -1.4089 | 0.00000  | 156.2500   |
| 4  | 4 | 2 | 1.2154  | 1.2154  | 0.00000  | 156.2500   |
| 5  | 4 | 4 | 1.0193  | 1.0193  | -0.00001 | 625.0000   |
| 5  | 5 | 5 | 0.9743  | 0.9743  | 0.00001  | 625.0000   |



***** Cálculo das grandezas do SEP *****



| i | Pi      | Qi      | Vi     | Thi      |
|---|---------|---------|--------|----------|
| 1 | 3.9484  | 1.1428  | 1.0000 | 0.0000   |
| 2 | -8.0000 | -2.8000 | 0.8338 | -22.4064 |
| 3 | 4.4000  | 2.9748  | 1.0500 | -0.5977  |
| 4 | -0.0002 | 0.0002  | 1.0193 | -2.8343  |
| 5 | 0.0002  | -0.0003 | 0.9743 | -4.5479  |


```

Figura 12 – Valores das grandezas do sistema-teste

Fonte: Autoria própria

Na Figura 12 são apresentados os resultados das variáveis definidas no modelo matemático. Note que o valor da função objetivo indica que não há presença de erros nas medidas e, portanto, o ponto de operação do sistema-teste tem sido estimado corretamente.

## CONCLUSÃO

Este artigo apresentou uma nova abordagem para o problema de estimação de estado em sistemas de potência utilizando um modelo de otimização matemática não linear. Neste trabalho, os resultados obtidos indicam que, a obtenção dos resultados através de um software de otimização matemática amplamente usado em aplicações na Engenharia, constitui-se em uma importante alternativa de solução para realizar estudos convencionais de estimação de estado em SEP.

A linguagem AMPL e o solver Knitro se mostraram eficientes no cálculo do ponto de operação do sistema-teste. Portanto, espera-se que este trabalho possa servir como referência para realizar estudos convencionais de estimação de estado em SEP.

Em trabalhos futuros, faz-se necessário incorporar no modelo matemático de otimização proposto um procedimento de tratamentos de erros que permita analisar o impacto de erros grosseiros no conjunto de medidas.

## REFERÊNCIAS

- EXPOSITO, A. G., ABUR, A. Power system state estimation: Theory and implementation. New York: Marcel Dekker, 2004.
- FAVERO, L. P.; BELFIORE, P. Pesquisa operacional para cursos de engenharia. Elsevier, 2013.
- FOURIER, R.; GAY, D. M.; KERNIGHAN, B. W. AMPL: A modeling language for mathematical programming. Thomson Publisher, 2003.
- GLOVER, J. D.; SARMA, M. S.; OVERBYE, T. J. Power system analysis and design. Ed. Cengage Learning, 2011.
- HOLMES, D. AMPL (A mathematical programming language) at the University of Michigan. Disponível em: <http://www-personal.umich.edu/~murty/510/ampl.pdf>. Acesso em: 20 mai. 2021.
- MONTICELLI, A. Electric power system state estimation. Proceedings of the IEEE, v.88, n.2, p. 262-282, 2000.
- MONTICELLI, A. Fluxo de carga em rede de energia elétrica. São Paulo, SP: Ed. Edgard Blücher, 1983.
- OLSZAK, A.; KARBOWSKI, A. Parampl: A simple tool for parallel and distributed execution of AMPL programs. **IEEE Access**, v. 6, p.49282-49291, 2018.